



# Gramática del proyecto OBLIQ

## Fundamentos de Lenguajes Programación

Carlos Andres Delgado S, Ing \*

Mayo de 2017

Algunas convenciones:

- Lo que va entre < > corresponde a un tipo de dato
- Lo que está entre comillas es texto que va en la estructura de la gramática
- La gramática inicia en <programa>

```
<programa> ::= <expresion>

<expresion> ::= <bool-expresion>
              ::= <identificador>
              ::= <numero>
              ::= <caracter>
              ::= <cadena>
              ::= ok
              ::= "var" { <identificador> = <expresion> }*(,) "in" <expresion> "end"
              ::= "let" { <identificador> = <expresion> }*(,) "in" <expresion> "end"
              ::= "let rec" { <identificador> "(" { <identificador> }*(,) ")" = <expresion> }* "in" <expresion> "end"
              ::= "set" <identificador> ":" <expresion>
              ::= "begin" <expresion> { <expresion> }*(,) "end"
              ::= <primitiva> "(" { <expresion> }* ")"
              ::= "if" <bool-expresion>
                  "then" <expresion>
                  { "elseif" <bool-expresion> "then" <expresion> }*
                  "else" <expresion>
                  "end"
              ::= "proc" "(" { <identificador> }*(,) ")" <expresion> "end"
              ::= "apply" <identificador> "(" { <expresion> }*(,) ")"
              ::= "meth" "(" <identificador> "," { <identificador> }*(,) ")" <expresion> "end"
              ::= "for" <identificador> "=" <expresion> "to" <expresion> "do" <expresion> "end"
              ::= "object" "{" { <identificador> "=" <expresion> }* "}"
              ::= "get" <identificador> "." <identificador>
              ::= "send" <identificador> "." <identificador> "(" { <identificador> }*(,) ")"
              ::= "update" <identificador> "." <identificador> ":" <expresion>
              ::= "clone" "(" <identificador> { <identificador> }*(,) ")"
              ::= <numero>

<bool-expresion>
  ::= true
  ::= false
  ::= <bool-primitiva> "(" { <expresion> }*(,) ")"
  ::= <bool-oper> "(" { <bool-expresion> }*(,) ")"

<primitiva> ::= +|-|*|%/&

<bool-primitiva> ::= < | > | <= | >= | is

<bool-oper> ::= not | and | or
```

---

\* carlos.andres.delgado@correounivalle.edu.co